/*
Shpaizman & Cavari
Minding (your own and) Others' Business: 
Assigning Co-Responsibility in Cabinet Decisions
Comparative Political Studies (likely 2023)

Data File:
Assigning Co-Responsibility in Cabinet Decisions.dta

Paper:
Shpaizman, Ilana, and Amnon Cavari. (year). "Minding (your own and) Others' Business: 
Assigning Co-Responsibility in Cabinet Decisions." Comparative Political Studies. 

Analysis with Stata 17
*/

* Figure 1: Number of Decisions per Year 
{
use "Assigning Co-Responsibility in Cabinet Decisions.dta", clear
graph bar, over(year, label(angle(90) labsize(small))) ///
	ylabel(,angle(0)) ytitle(Number of Decisions) ///
	scheme(s1mono)
}
	
* Table 1: Number of decisions by ministry and share of co-responsibility	
{
use "Assigning Co-Responsibility in Cabinet Decisions.dta", clear
asdoc tab Ministry_Code Coresponsibility, rowsort replace 
}

* Figure 2: Percent of decisions assigning co-responsibility over time
{
use "Assigning Co-Responsibility in Cabinet Decisions.dta", clear
recode Coresponsibility (1=1)(0=0), gen(yes)
recode Coresponsibility (0=1)(1=0), gen(not)
collapse (sum) yes not, by(year)
gen total=yes+not
gen percent_yes=100*yes/total
gen percent_not=100*not/total

lowess percent_yes year, scheme(s1mono) ///
	ytitle("Percent of Decisions") ylabel(0(25)100,angle(0)) yline(50) ///
	xtitle("") ///
	title("")
}


* Table 2: Assessing the Determinants of Assigning co-Responsibility 
{
use "Assigning Co-Responsibility in Cabinet Decisions.dta", clear

logit Coresponsibility ///
	BudgetAppropriation NewProgram /// 
	Saliency PartyLeader PartyPM z_gov_range_av ///
	Seatshare PM_Elect /// 
	i.majorIL 
	
logistic Coresponsibility ///
	BudgetAppropriation NewProgram /// 
	Saliency PartyLeader PartyPM z_gov_range_av ///
	Seatshare PM_Elect /// 
	i.majorIL

*robustness.  without co-responsibility decisions in which the minister of finance is co-responsible	
logit Coresponsibility ///
	BudgetAppropriation NewProgram /// 
	Saliency PartyLeader PartyPM z_gov_range_av ///
	Seatshare PM_Elect /// 
	i.majorIL ///
	if Coresponsibility==Coresponsibility_no_MOF 
}	
	
* Appendix C: Assessing the effect of ideological distance on 12 policy issues
{
use "Assigning Co-Responsibility in Cabinet Decisions.dta", clear

logit Coresponsibility ///
	BudgetAppropriation NewProgram /// 
	Saliency PartyLeader PartyPM  ///     
	Seatshare PM_Elect /// 
	CMPdifference 
}
	

